
Subject: Effects of Modular Coil Fields at S3 State on TF Coil Structural Continuity
1.0 Executive Summary
A 3D coupled electromagnetic-structural ANSYS
 model of the NCSX toroidal field (TF) and Modular Coil (MC) systems is assembled and used to determine Lorentz forces, deformations and stresses within the TF magnet from one particular equilibrium current set. The analysis focuses on the Modular Coil Fields at S3 state, when TF currents are low (B0 ~0.065 T), MC currents are high but negative in M1 and M3. Field analyses using current-sticks
 has shown that some of the TF coils have a positive net radial force of 6.1 kN (1400 lbf). This analysis is designed to check those earlier results [2] and draw a more detailed picture of the stress state.
The ANSYS model generally confirms the positive net radial forces acting on the TF coils, but predicts a smaller value (2.3 kN). This net outward load may be of little consequence compared to the complexities of the force distribution in the inner TF legs. The analysis shows that there is an average compressive toroidal stress of ~0.4 MPa from Lorentz forces and the reference outboard preload. This is sufficient, on-average, to develop a friction force (25 kN) which is much larger than the radial coil force (2-6 kN). The friction force would drop to 10 kN if not for the present 2000 lb per bracket preload. However, this approach appears to be too simplistic. 
A closer look at the interface stresses from a bonded model identifies non-uniformities in the stress field which include modest tensile stresses (would-be gaps) in some regions. Detailed analysis of the local compression and shear stresses on these wedged surfaces indicates that ~20% of the area could remain “stuck” while the other 80% produce small gaps or relative motion. Clearly, a nonlinear analysis with contact elements is required to accurately capture this behavior.
Simulations which include frictional contact between wedged faces of just two coils indicate that without the nominal 2000 lb per bracket preload, these faces will experience scissor-like relative motion with maximum slippage on the order of ±0.28 mm. When the 2000 lb per bracket preload is applied and a coefficient of friction (μ) of 0.3 is assumed, the adjacent coils hit hard against each other at the top and bottom of the interface and remain stuck. However, the average ratio of friction to normal forces is 0.27, indicating that there is little margin in the design if a μ of 0.3 is the design basis value. Intermediate locations separate, resulting in small (0.05 mm) coil-to-coil gaps. 
The stress in the side wall insulation peaks at these extremes of the interface, but remain an order of magnitude below the RT ultimate strength of a typical high-pressure laminate. Similarly, the maximum smeared winding pack stress is <20 MPa, which is well bellow the stress levels produced by more demanding load conditions
. 
In summary, the 2000 lb per bracket preload is just barely sufficient to maintain a no-slip condition in the TF coil system, at least for the subject equilibrium currents.

2.0 Assumptions and Notable Concerns
2.1 The stability of the TF coil system under light TF operating currents will rely on friction at the wedged surfaces to hold back the TF coils with a positive radial Lorentz force. An assumed friction coefficient of 0.3 is used in the analysis, which should be relatively easy to achieve. Most materials have friction coefficients greater than 0.3, and achieving a low friction surface usually takes some special effort. The subject coil currents push the minimum friction coefficient requirement to 0.27, which leaves little margin for other (if any) more demanding operating conditions. An increase in the design-basis coil-to-coil friction coefficient or preload is suggested.
2.2 The coil currents used here were originally chosen because some TF coils carry a small positive force tending to eject them coil from the coil system. However, the analysis shows that the detailed force distribution produces more of a poloidal torque with a small radial force superimposed. So, it is not clear that these coil currents represent the worst-case condition for producing slippage between coils. 

2.3 The computational demands of modeling six contact surfaces for this 120˚ field period model are too great for the resources available to Myatt Consulting, Inc. However, the EM analysis shows that a single contact surface in the middle of the six-coil model is the most likely place for slippage based on poloidal torque summations. The nonlinear structural analysis shows that this interface remains “stuck” during the reference equilibrium S3 field state. 

3.0 Analysis

The analysis is based on the 3D model shown in Fig. 3.0-1. There are six TF coils and six MCs which make up one 120˚ “field period.” Smeared TF coil winding packs (WP) are surrounded by ground-wrap insulation. Side-wall insulation separates the inner legs in the wedged region. 
The electromagnetic (EM) model is loaded by applied currents. The structural model is loaded by importing the Lorentz forces calculated by the EM model and the applied radial preload (located at ±52” about the equatorial plane). The model is very similar to the one described in [3] with the addition of MCs which are also fully described in a previous memo
. 
Deformations and stresses are postprocessed in order to assess the stability of the wedged surfaces (i.e., conditions which produce stick-slip behavior). Some results are supplemented with hand calculations for confirmation.

Fig. 3.0-1 Isometric View of One Field Period of TF & Modular Coils

For discussion purposes, we will refer to the six TF coils in the 120˚ model as TF1 through TF6, where TF1 is at θ=-50˚, TF2 is at θ=-30˚, TF3 is at θ=-10˚, TF4 is at θ=+10˚, TF5 is at θ=+30˚, TF6 is at θ=+50˚ and θ=0 is parallel to the global X axis.

(red), and nodal force vectors. Notice that for this particular EM loading, the preload forces are much larger than the Lorentz forces as denoted by their vector length. 
Fig. 3.0-2 Typical TF Coil with EM & Preload Force Vectors

4.0 Results

The results of the analyses are summarized in the sections below. Sect. 4.1 shows results for the case when only TF coils are energized (B0=0.5 T). This case is benchmarked against a previous analysis [3] for checking purposes. Sections 4.2 and 4.3 shows results for the subject loading (MC at S3 state and TF coils at 25.4 kA-t) when all wedged surfaces are bonded together (linear analysis), and when the inner coil pair have contact surfaces with friction and all other wedged surfaces bonded (nonlinear model). Within section 4.3, there are two analyses; one with a zero applied preload and a second with the nominal 2000 lb/bracket preload. This last section contains the most important findings, while the others are simply included to show the entire analysis chain and thought-process.
4.1 Model Benchmark, B0=0.5 T
When the TF coils are energized to 194.4 kA-t/coil, the on-axis toroidal field is 0.5 T. The parametric ANSYS model uses a single TF coil and 1/18th symmetry to simulate this operating condition.
Fig. 4.1-1 is a plot of the flux density in the smeared winding pack. The plot is annotated with various salient results. The peak field is reported to be 1.77 T and the “MX” symbol signifies that it appears just above equatorial plane. The magnitude of the field is perfectly consistent with 1/r scaling which would be 1.75 T at this location of the inboard leg.
The plot title also lists ∫JxBdV forces. The results are rounded to integers and all figures are reported. The net radial force acting on the coil is -373 kN which is ~5% higher than the -358 kN reported in [2]. The out-of-plane force sums to exactly zero, and the net vertical force is a mere 62 N which of course should be zero. These results produce some level of confidence that the model is reasonably accurate. 

Fig. 4.1-2 is a plot of the smeared stress intensity in the TF winding pack from this 0.5T operating condition. The results are included here for reference only, and show good agreement with results presented in an earlier memo [3].
4.1-1 Flux Density in Smeared TF Coil Winding Pack, 0.5 T Operating Condition
[image: image3.png]

Fig. 4.1-2 Stress Intensity in Smeared Winding Pack, 0.5 T Operating Condition
4.2 Modular Coil at S3 state, TF coils at 25.4 kA-t (Bonded Inner Legs)
When fields from the MCs are included in the analysis the field period increases to 120˚. Now, the model must include 6 TF coils, 6 MCs and invoke 1/3rd electromagnetic symmetry. 
Fig. 4.2-1 is a plot of the flux density in the TFCs and MCs from the subject operating condition; total currents are embedded in the plot title. The maximum field of 4.7 T occurs in MC3. It is worth noting that the MCs are modeled with a rather coarse mesh, which may not be sufficient to capture the actual peak field. However, the element density is probably sufficient to serve as current sources for the Biot-Savart field calculation within the denser TF winding pack mesh.
Fig. 4.2-2 is a plot of the force density in TF coil #3 (θ=-10˚). Again, net forces are embedded in the plot title, which shows a 2.3 kN net radial force acting on this (and the θ=+10˚) coil. The radial forces acting on three typical TF coils are listed below.
ANSYS Solid Model [kN]: Fr(TF1)=-4.5
Fr(TF2)=-14.7

Fr(TF3)=+2.3



Brooks’ Stick Model [kN]: Fr(TF1)=-3.4
Fr(TF2)=-17.3

Fr(TF3)=+6.1


The two models produce qualitatively similar results. On a percentage basis, there is some disparity, particularly with TF3. On an absolute value basis, the average disparity is about 2.5 kN (~600 lb). One might expect accuracy in the ANSYS model to be on the order of 100 N (based on its ability to produce a near zero net vertical force from the TF-only case, Fig. 4.1-1). However, only a mesh refinement or another code would provide a reasonable second benchmark. The present 270,000 node model has a 6+ hour run-time which makes a mesh-refinement study more of a luxury than a necessity. 
The model is also used in a linear structural analysis. Magnetic elements are replaced with structural elements, boundary conditions are applied and Lorentz forces are imported from the field analysis. 
The EM loads associated with this reference case produce a non-symmetric loading in the toroidally continuous inboard legs. Fig. 4.2-3 is a view looking down the coil system axis (Z). Toroidal stresses are superimposed on the greatly exaggerated model deformations. EM forces drive some portions of the arc into “flats” and other portions into “tight bends.” Of course these terms are loosely used to illustrate the nature of the deformations as no cross-section through the six-coil inner leg region actually becomes flat or tightly bent. That said both of these deformations lead to tension on one side of the WP and compression on the other side, superimposed on an average distribution. It’s worth noting that these toroidal stresses are generally very light, typically ±1 MPa.
Another way to illustrate this result is shown in Fig. 4.2-4. In this case the contours are binary, highlighting regions of tension (grey) and compression (red). Look closely at the right (or left) edge of the right contour plot. This area represents the contact surface between this 120˚ sector and the next. The grey (tension) areas would pull apart unless the inner legs were bonded at assembly. Instead, small (tiny) gaps will open up between the adjacent TF sidewall insulation layers. An import question remains unanswered by this linear analysis: Will the gaps be extensive enough to lead to relative motion and irreversible changes in the coil positions? Before jumping to the obvious path to answer this question (i.e., a nonlinear model with contact elements on the surfaces of adjacent inner TF legs), let’s do some simple calculations.
Let’s calculate the average toroidal stress, σθ, on the wedged surfaces. This stress is simply the total force normal to the wedged surface (F() divided by the contact surface (Awedged). In this analysis, the normal force can be obtained by summing the average magnetic force (from the three typical coils) and the applied preload force and then dividing by the sine of the wedge half-angle (10˚). For simplicity, we’ll use half of the coil force and react it over one wedged face.
σθ 
= {(Fmagnetic) + (Fpreload)}/(sin10˚)/(Awedged) 

= {(-4500 N – 14700 N + 2300 N)/3 + (-2000 lb/0.2245 lb/N)}/( sin10˚)/(0.1804 m2)
= {(-5600) + (-8900)}/( sin10˚)/(0.1804 m2)

= -0.46 MPa

From the 3D analysis, when the stress normal to the side wall insulation is averaged over the entire wedged surface the average toroidal stress is -0.37 MPa. This is in reasonably good agreement with -0.46 MPa from the hand-calculation. Its worth noting, however, that the hand calculation must be correct. The difference between the two values is probably due to finite element effects (i.e., steep stress gradients over a coarse mesh). 
Maintaining an average toroidal compression, despite the positive radial EM force on six of 18 coils, does not guarantee a slip-free structure. Another analysis approach would be to multiply the average toroidal compressive stress by a friction coefficient (μ) and surface area to obtain the nominal radial force which can be carried by friction:
F(friction) = μ σθ Awedged = (~0.3) (0.1804 m2) (0.46 MPa) = 25 kN
Since 25 kN is greater than both 2.3 kN and 6.1 kN, it would appear as if friction on the wedged faces is sufficient to keep these coils with positive radial force stuck to their neighbors. It is interesting to note that more than 60% (8900/14500) of the average toroidal compression is a result of the preload (2000 lb/bracket x 2 brackets per TF coil). If the preload is omitted, the friction force drops to 10 kN, which is far more marginal (depending on which radial force is correct). Also, when the TF coils carry their full (0.5 T) current, the radial force per coil is ~373 kN (84,000 lb) or 21 times the mechanical preload. So, the preload is only of value at these very low TF coil currents.
Yet another approach to analyzing the stability of the coil-to-coil interfaces is to examine normal and shear stresses. This is a far more accurate measure of the wedging surface characteristics since it uses local shear and normal stresses from the model and not surface-wide average values. The analysis goes as follows. At each inner leg side-wall surface, the local shear stress is subtracted from a friction coefficient multiplied by the normal stress to create the so-called Shear Margin. 

Shear Margin = Shear Capacity – Shear Stress = (-μσθ) - (τθz2+τrθ2)0.5
A minus sign in front of the μσθ term converts toroidal compressive stresses into positive shear capacity and toroidal tensile stresses into negative shear capacity. Values of Shear Margin which are greater than zero signify that local friction is greater than local shear.

This methodology is applied to the six wedging interfaces in the model. However, only one set of normal, shear and shear margin stress contour plots are presented in plot form. 
Fig. 4.2-5 is a plot of the normal (toroidal) stress in the sidewall insulation located between TF3 & TF4. Contours are limited to small negative values which helps illustrate the nominal compressive stress distribution. The plot shows that the middle portion of the inner leg sees compressive stresses of about 0.2 MPa. Immediately above and below this middle region, the normal stresses become mildly tensile, with a maximum value reported to be ~3 MPa. A narrow band of contours close to the extremes of the wedged interface signify a sharp transition from these modest tensile values to large compressive values at the tips. Since the model assumes a bonded interface, these tensile values are “false” and the compressive stress is overestimated. The ANSYS plot title captures the integrated average normal stress: -0.36 MPa. 

Even without bonding, a compressive normal stress and friction could support shear stress at the wedged interface. However, the capacity of the surface to carry a shear stress without slipping will be a function of the local compression, shear and friction coefficient. As long as the product of the compressive stress and friction coefficient is greater than the local shear stress, the interface will remain “stuck.” Fig. 4.2-6 is a plot of the shear stress modulus (τrθ2 + τθz2)0.5 on this particular sidewall insulation layer. Again, contour values are carefully chosen to provide the most insight into the distribution. Notice that the shear stress peaks at the tips of the interface, but is generally <0.2 MPa over the bulk of the area. The plot title also captures the average shear: 0.15 MPa.
Aside: If these stresses were uniform, then a friction coefficient of at least 0.15/0.36 or 0.42 would be required to maintain a stuck interface.

Now we can combine these two results. Shear stresses are subtracted from the normal compressive stress multiplied by a friction coefficient. Positive values represent a shear margin or excess frictional capacity relative to shear stresses. Fig. 4.3-7 shows two plots which represent the shear margin assuming friction coefficients of 0.3 (top) and 0.6 (bottom). The titles list the area fraction in which the shear margin is a positive value: 0.16 for μ=0.3 and 0.24 for μ=0.6. 
A review of the other five interfaces indicates that a positive shear margin value occurs over 15% to 29% of the wedged interfaces. So, this θ=0 interface represents one of the worst locations. In any case, the results show that most of the interface areas do not have sufficiently high compression to support the shear stresses developed by the coil deformations. This means that local movement between TF coils is likely.
This approach is somewhat more convincing than the previous two analyses which are based on average surface conditions. However, this bonded interface model overestimates the magnitude and extent of toroidal compression which are artificially augmented by tensile stresses that cannot develop. A more rigorous approach would be to model general contact surfaces at the wedging interfaces and look for relative motion. This is complicated by the importance of friction, which makes the analysis not only nonlinear, but path-dependent. The result is a very long computer run. 

Fig. 4.2-1 Flux Density in TFCs & MCs from 0.065T, S3 Field State

Fig. 4.2-2 Force Density in TF Coil #3 (θ=-10˚), TFCs & MCs from 0.065T, S3 Field State

Fig. 4.2-3 Toroidal Stress, Inner Legs, TFCs & MCs from 0.065T, S3 Field State

Fig. 4.2-4 Binary Toroidal Stress Contours, Inner Legs, TFCs & MCs from 0.065T, S3 Field State
Fig. 4.2-5 Normal Stress Contours Typical Wedged Surface

TFCs & MCs from 0.065T, S3 Field State

Fig. 4.2-6 Shear Stress Contours Typical Wedged Surface

TFCs & MCs from 0.065T, S3 Field State

Fig. 4.2-7 Shear Margin Contours Typical Wedged Surface (top: μ=0.3, bottom: μ=0.6)

TFCs & MCs from 0.065T, S3 Field State

4.3 Modular Coil at S3 state, TF coils at 25.4 kA-t (Friction at TF3/4 Contact Surface)

The linear (no-slip) analysis presented above is inherently unconservative. Artificial toroidal tensile stresses develop in certain regions (recall Figs. 4.2-4 & 4.2-5) and lead to artificially high toroidal compression in other regions. In an effort to perform a more representative analysis, node-to-node contact elements are added to the wedged faces of the inner legs. However, this approach resulted in a multi-day computer run that showed no signs of converging and was terminated before completion. As a compromise, the model is reworked to reduce the number of elements by eliminating the ground wrap elements, and limiting the node-to-node contact to the single TF3/4 interface. This simplified model is shown in Fig. 4.3-1 and is again used to evaluate the EM forces for the reference equilibrium currents. However in this case, the “stuck” or “sliding” result will appear in a simple deformation plot, not postulated from postprocessing shears and normal stresses.

Fig. 4.3-1 Simplified TF Coil Model, Ground Wrap Eliminated, Contact at A-A Interface Only

Zero Preload

Fig. 4.3-2 is a plot of the radial displacement contours on a greatly exaggerated deformed coil. There is no radial preload in this analysis. Notice the following: 

· the outboard legs of Type-A and B coils tend to separate (pull away from the Y=0 plane)

· the –Y coils tend to move radially in at the top (and out at the bottom)
· the +Y oils tend to move radially out at the top (and in at the bottom)

These displacement trends are consistent with the linear (fully bonded) analysis reported in the previous section as shown in Fig. 4.2-3. Unlike the bonded configuration, the TF3/4 interface can demonstrate relative motion or scuffing as a result of these EM loads. Relative motion at the contact surface is postprocessed and determined to be 0.55 mm at the extremes of the inner legs. That is, the top of TF3 moves radially inward about 0.28 mm while the top of the TF4 moves radially outward about 0.28 mm. This particular analysis does not have a radial preload, which is evident by the integrated toroidal compression listed in the plot title: -0.08 MPa. Analysis of the other interfaces shows similar results, with average toroidal stresses of ~0.1 MPa. Recall that the toroidal compression is 0.36 MPa in the linear model with 2000 lb/bracket preload.

Fig. 4.3-2 Radial Displacements, TFCs & MCs from 0.065T S3 Field State

(μ=0.3 Friction at A-A surface)

SAFJ

2000 lb Preload

Fig. 4.3-3 is a plot of the loaded coil system with greatly exaggerated deformations (300x). In this case, a 2000 lb/bracket radial preload is applied. That’s 4000 lb per coil with preload hardware located 52” above and below the equatorial plane. 

The plot shows that the preload causes the coils to hit hard against each other at the top and bottom of the contact surface (TF3/4 interface) and open a gap in between. With a friction coefficient of 0.3, this analysis shows that the coils do not move relative to each other as they did when only low-level EM forces provided a 0.08 MPa toroidal interface pressure. Here, the EM and preload combine to produce an average toroidal pressure of 0.36 MPa, as shown in the ANSYS plot title. The largest gap is reported to be ~0.05 mm.

Fig. 4.3-3 Deformations in the Preloaded TF Coil System, TFCs & MCs from 0.065T S3 Field State (μ=0.3 Friction at TF3/4 interface)
[image: image16.png]TA\ANSYS Mechanical/Emag Utility Menu (tfmodb10)

File Select List Plot PlotCtls WorkPlane —Parameters Macro MenuChrls  Help
SEEIEIEE R

ANSYS Toobar

save ] pesu o] qur| powneen] peptor] oemase| enase| o ue |

[ e )|
E Preferences ANSYS 8.1
Flux Density, Only TF Coils Energized to 0.5 T JUL 12004
B 11:35:10
£ Data & Fle Opts
NODAL SOLUTION
STEP=2
SUB =1
TIME=2

Query Results

Breeas s BSUM _(AVG)
n Write PGR File SMN =.311E-03
Bmax = 0.5 (1.4/0.4) SM 69

=175T .311E-03

196847
1393382
589917
786453
982988
118
oo mogostice 1376
- 1.573
1.769

Net Force Summation

5 Session Editor tfmodb10, TF#1 (0 deg),|Local FX/FY/FZ=-373328/0/62 N|

[ Finish

EEBERErREEEEEbrEEEEREE

Pick 15t corner = type=1 real=1 coys=1



As shown in Fig. 4.3-4, the average surface toroidal stress is composed of a “high” contact stress (~32 MPa) at the extremes and zero in between.

Fig. 4.3-4 Wedging Pressure in SW Insulation at TF3/4 Interface

TFCs & MCs from 0.065T S3 Field State (μ=0.3 Friction at TF3/4 surface)
Fig. 4.3-5 is a plot of the relative motion at the TF3/4 contact surface. The contour values represent the amount of relative motion and indicate a maximum of 27 μm or 1 mil. This maximum occurs in the red regions slightly above and below the Z=0 equatorial plane. Sliding at the extremes of the interface is nil, as represented by the “MN” symbol which correspond to slip of 10-20 m.
The plot includes the deformed shape. Nodes on one side of the interface which move relative to nodes on the other side of the interface trace the motion relative to the unloaded position. This plot shows how the inboard legs are bent into an “S” with one side going in one direction while the other side goes in the opposite direction. 

The ANSYS plot title also lists some interesting results about the status of the contact element: 2.6% are stuck, 2.6% are sliding, and 94.8% are open. In addition, of the contact elements which are closed (i.e., in contact and sliding or stuck) the average ratio of friction force to normal force is 0.27. This represents the friction coefficient required to maintain these displacements. 

Fig. 4.3-5 Relative Motion at TF3/4 Interface

TFCs & MCs from 0.065T S3 Field State (μ=0.3 Friction at TF3/4 surface)
Fig. 4.3-6 is included to show the stress intensity in the smeared WP as a result of this equilibrium plus mechanical preload operating condition. The plot shows that the highest stresses appear at the lateral displacement boundary conditions, which make the 32 MPa value suspect. Away from these lateral constraints, the plot indicates larger regions at a stress level of 15-20 MPa (in the transition region of the TF2 & TF5 coils). Even with the understanding that these are smeared stresses which must be “marked-up” to determine the stress in the Cu conductor, the magnitude is probably low enough to not be a concern.

Fig. 4.3-6 Stress Intensity in the Smeared TF WP, TFCs & MCs from 0.065T S3 Field State
Is this the correct interface to model as a stick-slip plane?
As we have already discussed, computational limitations have lead to the use of a single stick-slip plane at θ=0˚ in this nonlinear analysis. It might seems apparent from Figs. 4.2-3 and 4.3-2 that there is a general twist in which the three –θ coils (TF1, 2 & 3) tend to rotate in one poloidal direction while the three +θ coils (TF4, 5 & 6) tend to rotate in the opposite poloidal direction. Fig. 4.3-2 illustrates this best since slippage occurs and the two sets of coils are displaced relative to each other. We will call this a poloidal twist, and the load which drives these displacements a poloidal torque.
It is not [yet] clear that the critical slip plane has been chosen to be analyzed. It is possible, for example, that poloidal torques on TF1 & TF3 are opposite to TF2, and their sum is a small negative value. Likewise, the poloidal torques on TF4 & TF6 are opposite to TF5, and their sum is a small positive value. If this were the case, then stick-slip planes on either side of TF2 or TF5 would be more important to model since this would be the location of the largest torque sign change.
A poloidal torque (Tθ) is calculated for each coil relative to a local coordinate system aligned with the coil mid-plane (i.e., at θ=-50˚, -30˚, -10˚, and so on):

Tθ = Fx Z + Fz(R0-X)
where R0 is the major radius, X is the radial in-plane coordinate, Z is the vertical in-plane coordinate, and Z=0 is the equatorial plane of the TF magnet system. Fx & Fz are Lorentz forces ∫JxBdV from the EM analysis. Table 4.3-1 lists the poloidal torques for each of the six TF coils. There are a few things to notice from these results:
· Stellarator-symmetry is evident (Tθ1 = -Tθ6, Tθ2 = -Tθ5 Tθ3 = -Tθ4)

· The EM calculation produces consistent results (Tθ1 ≈ -Tθ6, Tθ2 ≈ -Tθ5 Tθ3 ≈ -Tθ4)
· Most importantly, coils 1, 2 & 3 torques carry the same sign

The fortunate conclusion is that the stick-slip plane separating TF3 and TF4 is where the maximum torque gradient appears. If the slip is to occur from this EM loading, it will happen at this interface. The analysis with μ=0.3 on this interface shows that no slip occurs and that the coils remain “stuck.”
Table 4.3-1 Summary of Poloidal Torque Values (about R0)

	Coil Number
	Poloidal Torque Summation

[N-m]

	1
	-3234

	2
	-6004

	3
	-2596

	4
	+2608

	5
	+6017

	6
	+3247


5.0 Summary & Commentary
The EM and linear structural model indicate that the conditions at the TF inner leg wedged surfaces are a strong function of the local EM force distribution created by TF and MC current sources. An analysis of normal and shear stresses at these surfaces in the linear model indicates that <30% of the area might remain “stuck” due to friction. However, toroidal tensile stresses are an artifact of the linear model, and to some extent, augment the toroidal compression producing nonconservative results. That is, there is less toroidal compression in the unbonded coil system than predicted by the linear model. 

A nonlinear analysis with frictional contact elements is presented. When no preload is applied, shear stresses generally exceed the frictional capacity of the wedged surfaces resulting in relative motion between coils. The scuffing is on the order of ±0.28 mm at the top and bottom of the interface. When a 2000 lb/bracket preload is applied, high localized wedging pressures develop and the coils stick without sliding. Contact stresses in the sidewall insulation and smeared stresses in the WP appear to be well below other operating conditions.

Computational limitations restrict the model to a single unbonded surface between coils TF3 & TF4. A torque summation indicates that this is indeed the location most susceptible to slippage as all of poloidal torque from TF1, 2 & 3 coils oppose the poloidal torque from TF4, TF5 & TF6. 
It is worth noting that while the reference EM load case produces the largest positive radial force on a TF coil [2], it is not certain that this current scenario produces the worst condition for slippage. There may be another load case which could result in vertical slippage even though the preload is well-sized to handle radial slippage. 
6.0 Attachments

6.1 Drawings
6.1 ANSYS Input File

/batch

rn=1

/filnam,tfmodb2%rn%

/show,tfmodb2%rn%a,grp

!/filnam,tfmods2%rn%

resume

*if,1,eq,1,:1000

/prep7

/com

/com Electromagnetic analysis of the NCSX TF and Modular Coil System

/com Structural analysis of the TF Coil System (No OOP structure...just inboard wedging and BCs)

/com Geometry defined by: se131-092, se131-091, tf_cond

/com Solve the Conduction and field problem, then solve the structural problem

/com Derived from tfmod1.dat, Coils made as smeared without ground wrap or sidewall insulation.

/com This coarser model designed to run with wedging contact elements to determine potential for stick-slip

/com 

/com

/com Model by Leonard Myatt, Myatt Consulting, Inc.

/com   leonard.myatt@myattconsulting.com, 508-520-4590

/com

/com

/com Run History

/com

/com 20: TF Only...looks OK

/com 21: EM Only: TF & MC currents from Art Brooks (TF:+25.4 kA-t, M1/M2/M3:-850/+850/-850 kA-t)

/com

/com

/com Contact and Mesh Density Parameters

k_nl=1                   ! 0: Wedged surfaces held to UY=0, 1: Generate Flex-Rigid Contact Interface

k_dens=3                 ! number of elements across WP (5 is good, smaller can cause meshing problems)

mu_sw=0.3                ! sidewall-to-sidewall friction coefficient

/com Misc Parameters

k=0.0254                 ! english to si conversion factor

pi=acos(-1)              ! pi

mu0=4*pi*1e-7            ! Mu0

*afun,deg                ! use degrees in trig functions

t=0.0001                 ! a tiny length

th=0.1                   ! a tiny angle

/com Nominal Performance Parameters

f_preload=2000/0.2248    ! applied radial preload per TF Bracket x 2 brackets per coil

z_preload=52*k           ! vertical location of applied preload

b0=0.5                   ! nominal flux density (one way of specifying TF current)

r0=1.4                   ! major radius

ntf=18                   ! number of tf coils

nsym=3                   ! toroidal symmetry

arc=360/ntf              ! angular extent of 1 tf coil

emsym,nsym               ! electromagnetic symmetry about Z

dtmp=0!-215                ! differential temperature (WRT room temp)

/com

/com Coil Currents

/com

i_tf=25.4e3 !5e6*b0*r0/ntf       ! total current in each smeared tf winding pack, A-turns/coil

i_mc1=-42500             ! conductor current in MC1, Amps

i_mc2=+42500             ! conductor current in MC2, Amps

i_mc3=-47222             ! conductor current in MC3, Amps

/com

/com element types

/com

et,1,98,2             ! Insulation (UX, UY, UZ)

et,2, 5,1             ! Modular Coil WP (TEMP, VOLT, MAG)

et,3,98,1             ! TF WP (TEMP, VOLT, MAG)

/com

/com graphics keys

/com

/pnum,mat,1

/num,1

/dist

/focus

/vup,1,z

/view,1,,-1

/com

/com tf conductor, ignoring radii (in local CS vernacular)

/com

n_lay=6                              ! number of radial layers

n_pan=2                              ! number of pancakes

dr_con=0.434*k                       ! Conductor build in thickness

dz_con=2.513*k                       ! Conductor build in height

dri_con=0.190*k                      ! inside thickness of cooling channel

dzi_con=1.130*k                      ! inside height of cooling channel

a_con=dr_con*dz_con-dri_con*dzi_con  ! conductor metal area

/com

/com Insulation

/com

t_tw=0.049*k              ! turn wrap insulation thk

t_pan=0.03*k              ! pancake insulation thk (can be 0.0)

t_lay=0.0*k               ! layer insulation thk (can be 0.0)

t_gw=0.12*k               ! module over-wrap thickness

t_sw=0.16*k               ! sidewall insulation thickness

t_gap=t_gw/10             ! height of gap in outboard leg for load application

/com

/com WP Build

/com

dr_wp=n_lay*(dr_con+2*t_tw)+(n_lay-1)*t_lay   ! radial build of WP

dz_wp=n_pan*(dz_con+2*t_tw)+(n_pan-1)*t_pan   ! toroidal build of WP

dr_iwp=dr_wp+2*t_gw                           ! radial build of ground-wrapped WP

dz_iwp=dz_wp+2*t_gw                           ! toroidal build of ground-wrapped WP

cel=sqrt(dr_wp**2+dz_wp**2)/k_dens            ! characteristic element size

/com

/com Centers of sweep and inside radii

/com

x11=r0                       ! x position of local 11

z11=0.0                      ! z position of local 11

ri11=r0-(12.494*k+dr_iwp)    ! inside radius of coil in local 11

/com

x12=43.239*k                 ! x position of local 12

z12=28.785*k                 ! z position of local 12

ri12=27.314*k                ! inside radius of coil in local 12

/com

x13=53.629*k                 ! x position of local 13

z13=7.141*k                  ! z position of local 13

ri13=51.323*k                ! inside radius of coil in local 13

/com

x14=53.629*k                 ! x position of local 14

z14=0.000*k                  ! z position of local 14

ri14=ri13+z13                ! inside radius of coil in local 14

/com

x15=58.111*k                 ! x position of local 15

z15=10.850*k                 ! z position of local 15

ri15=47.614*k                ! inside radius of coil in local 15

/com

ri16=199.614*k               ! inside radius of coil in local 16

x16=(97.051+12.494-203.046)*k! x position of local 16

z16=0.000*k                  ! z position of local 16

/com

x17=58.111*k                 ! x position of local 17

z17=-10.850*k                ! z position of local 17

ri17=47.614*k                ! inside radius of coil in local 17

/com

x18=58.111*k                 ! x position of local 18

z18=0.000*k                  ! z position of local 18

ri18=ri14                    ! inside radius of coil in local 18

/com

x19=53.629*k                 ! x position of local 19

z19=-7.141*k                 ! z position of local 19

ri19=51.323*k                ! inside radius of coil in local 19

/com

x20=43.239*k                 ! x position of local 20

z20=-28.785*k                ! z position of local 20

ri20=27.314*k                ! inside radius of coil in local 20

/com

/com Swept Angles (and straight-lengths)

/com

dy11=2*z12

dy12=abs(atan((z12-z13)/(x13-x12)))

dy13=90-dy12

dy14=4.482*k                 ! length of section 14

dy15=90-abs(atan((z15-z16)/(x15-x16)))

dy16=2*abs(atan((z15-z16)/(x15-x16)))

dy17=dy15

dy18=dy14

dy19=dy13

dy20=dy12

/com

/com material properties

/com

/com TF Coils

mp, kxx,2,1

mp,murx,2,1

mp,rsvx,2,1

mp,alpx,2,14e-6    ! roughly (no reference or calc)

mp,  ex,2,125E9

mp, Gxy,2, 45E9

mp,nuxy,2,0.39

/com

/com Side Wall insulation (Y is through thickness)

/com

mp,murx,3,1

mp,  ex,3,24e9

mp,alpx,3,8e-6

mp, gxy,3,10e9

mp,nuxy,3,0.2

mp, kxx,3,1

/com

/com coordinate systems

/com

/com Local 11

csys

wpcsys

wpoff,x11,,z11

wprot,,-90

wprot,180

cswpla,11

/com Local 12

csys

wpcsys

wpoff,x12,,z12

wprot,,-90

wprot,180

cswpla,12,1

/com Local 13

csys

wpcsys

wpoff,x13,,z13

wprot,,-90

wprot,180

wprot,dy12

cswpla,13,1

/com Local 14

csys

wpcsys

wpoff,x14,,z14

wprot,,-90

wprot,180

wprot,90

cswpla,14

/com Local 15

dy1215=90

csys

wpcsys

wpoff,x15,,z15

wprot,,-90

wprot,180

wprot,90

cswpla,15,1

/com Local 16

csys

wpcsys

wpoff,x16,,z16

wprot,,-90

wprot,180

wprot,90+dy15

cswpla,16,1

/com Local 17

csys

wpcsys

wpoff,x17,,z17

wprot,,-90

wprot,180

wprot,180+dy16/2

cswpla,17,1

/com Local 18

csys

wpcsys

wpoff,x18,,z18

wprot,,-90

wprot,180

wprot,180+90

cswpla,18

/com Local 19

csys

wpcsys

wpoff,x19,,z19

wprot,,-90

wprot,180

wprot,180+90

cswpla,19,1

/com Local 20

csys

wpcsys

wpoff,x20,,z20

wprot,,-90

wprot,180

wprot,180+90+dy19

cswpla,20,1

/com

/com Smeared Winding Pack

/com

/com Straight Leg

csys,11

wpcsys

block,ri11,ri11+dr_iwp,-dy11/2,,-dz_iwp/2,dz_iwp/2

!block,ri11+t_gw,ri11+dr_iwp-t_gw,-dy11/2,,-dz_wp/2,dz_wp/2

!vovlap,all

block,ri11,ri11+dr_iwp,,dy11/2,-dz_iwp/2,dz_iwp/2

!block,ri11+t_gw,ri11+dr_iwp-t_gw,,dy11/2,-dz_wp/2,dz_wp/2

!vovlap,all

/com Local 12,13 Arc

*do,j,12,13

csys,j

wpcsys

ksel,s,loc,x,0.99*ri%j%,1.01*ri%j%

*get,ri%j%,kp,,mnloc,x

vsel,none

cylind,ri%j%,ri%j%+dr_iwp,-dz_iwp/2,dz_iwp/2,,dy%j%

!cylind,ri%j%+t_gw,ri%j%+dr_iwp-t_gw,-dz_wp/2,dz_wp/2,,dy%j%

!vovlap,all

*enddo

/com Top Straight

csys,14

wpcsys

ksel,s,loc,x,0.99*ri14,1.01*ri14

*get,ri14,kp,,mnloc,x

vsel,none

block,ri14,ri14+dr_iwp,,dy14,-dz_iwp/2,dz_iwp/2

!block,ri14+t_gw,ri14+dr_iwp-t_gw,,dy14,-dz_wp/2,dz_wp/2

!vovlap,all

/com Local 15,16,17 Arc

*do,j,15,17

csys,j

wpcsys

ksel,s,loc,x,0.99*ri%j%,1.01*ri%j%

*get,ri%j%,kp,,mnloc,x

vsel,none

cylind,ri%j%,ri%j%+dr_iwp,-dz_iwp/2,dz_iwp/2,,dy%j%

!cylind,ri%j%+t_gw,ri%j%+dr_iwp-t_gw,-dz_wp/2,dz_wp/2,,dy%j%

!vovlap,all

*enddo

/com Bot Straight

csys,18

wpcsys

ksel,s,loc,x,0.99*ri18,1.01*ri18

*get,ri18,kp,,mnloc,x

vsel,none

block,ri18,ri18+dr_iwp,,dy18,-dz_iwp/2,dz_iwp/2

!block,ri18+t_gw,ri18+dr_iwp-t_gw,,dy18,-dz_wp/2,dz_wp/2

!vovlap,all

/com Local 19,20 Arc

*do,j,20,19,-1

csys,j

wpcsys

ksel,s,loc,x,0.99*ri%j%,1.01*ri%j%

*get,ri%j%,kp,,mnloc,x

vsel,none

cylind,ri%j%,ri%j%+dr_iwp,-dz_iwp/2,dz_iwp/2,,dy%j%

!cylind,ri%j%+t_gw,ri%j%+dr_iwp-t_gw,-dz_wp/2,dz_wp/2,,dy%j%

!vovlap,all

*enddo

/com

/com Glue Up WPs

/com

allsel

vglue,all

/com

/com Trim to make wedged region

/com

allsel

csys

*get,zmx,kp,,mxloc,z

ksel,s,loc,x,,x12

lslk,,1

asll,,1

vsla,,1

cm,tf,volu

csys

wpcsys

wprot,arc/2

wpoff,,-t_sw

cswpla,21

block,.01,r0,,r0,-1.5*zmx,1.5*zmx

csys

wpcsys

wprot,-arc/2

wpoff,,+t_sw

cswpla,22

block,.01,r0,-r0,,-1.5*zmx,1.5*zmx

btol,t_tw/20

vsbv,tf,all

btol

/com

/com Add Wedged-Region Insulation

/com

csys,21

asel,s,loc,y

vext,all,,,,t_sw

csys,22

asel,s,loc,y

vext,all,,,,-t_sw

/com

/com Slit Outboard Leg for Volt BC

/com

csys

wpcsys

allsel

cm,tf,volu

block,r0,3*r0,-dz_wp,dz_wp,,t_gap

btol,t_gap/2

vsbv,tf,all

/com

/com Cut Model in half for reflection later

/com

allsel

csys

wpcsys

cm,tfcoil1,volu

block,,3*r0,-dz_wp,,-10,10

vsbv,tfcoil1,all

/com

/com Set Attributes

/com

csys

ksel,s,loc,y

lslk

asll

vsla

vatt,2,,3

/com Attributes of SW Insulation

vsel,s,mat,,0

vatt,3,,1

/com

/com Mesh the smeared WP Volumes

/com

MSHAPE,1,3D 

/com Insulation (could be improved)

esize,cel

vsel,s,mat,,3

vmesh,all

esize,cel

vsel,s,mat,,2

vmesh,all

/com

/com Reflect other half

/com

allsel

*get,dn1,node,,num,max

csys

nsym,y,dn1,all

esym, ,dn1,all

numm,node

/com

/com Make all TF Coils

/com

allsel

*get,dn2,node,,num,max

csys,1

ngen,ntf/nsym,dn2,all,,,,360/ntf

egen,ntf/nsym,dn2,all

*get,thmn,node,,mnloc,y

*get,thmx,node,,mxloc,y

modmesh,detach

ngen,2,,all,,,,-(thmn+thmx)/2

*get,thmn,node,,mnloc,y

*get,thmx,node,,mxloc,y

/com Fix ESYS in Straight Leg Wedged Faces

esel,s,mat,,3

local,101,1

emodif,all,esys,101

csys

/com

/com Couple & Ground Volts and Apply Current to TF WP

/com

*do,j,1,ntf/nsym

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,x,r0,12

nsel,r,loc,z

nsel,r,loc,y,-(360/nsym/2)+(360/ntf)*(j-1),-(360/nsym/2)+(360/ntf)*(j-0)

cp,next,volt,all

*get,n_tf%j%,node,,num,min

f,n_tf%j%,amps,i_tf

/com Ground other end

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,x,r0,12

nsel,r,loc,z,t_gap

nsel,r,loc,y,-(360/nsym/2)+(360/ntf)*(j-1),-(360/nsym/2)+(360/ntf)*(j-0)

d,all,volt

*enddo

/com

/com Modular Coil Model

/com

/input,modcoils1,cdb     ! pull in PDR modular coils from HM Fan's file-1.7t.db

n_mcoils=3               ! number of Modular Coil Pairs

tpmc1=10                 ! turns per MC1 winding pack, 2 WP per coil, 4 in-hand

tpmc2=10                 ! turns per MC2 winding pack, 2 WP per coil, 4 in-hand

tpmc3=9                  ! turns per MC3 winding pack, 2 WP per coil, 4 in-hand

/com Node numbers for current application

n_m11=13599

n_m12=13609

n_m13=13619

n_m14=15619

n_m21=21639

n_m22=21649

n_m23=21659

n_m24=23659

n_m31=29679

n_m32=29689

n_m33=29699

n_m34=31699

/com Make nodal components out of Modular Coil Current Nodes

*do,jj,1,n_mcoils

nsel,none

*do,j,1,4

nsel,a,node,,n_m%jj%%j%

*enddo

cm,n_m%jj%,node

*enddo

/com

/com Modular Coil Currents

/com

cmsel,s,n_m1 $f,all,amps,i_mc1*tpmc1       ! MC1

cmsel,s,n_m2 $f,all,amps,i_mc2*tpmc2       ! MC2

cmsel,s,n_m3 $f,all,amps,i_mc3*tpmc3       ! MC3

/com Nix MAG DOF

esel,s,type,,2,3

nsle

d,all,mag

/com Set TEMP

d,all,temp,dtmp

allsel

bfunif,temp,dtmp

allsel

/psym,csys

nit=nint(i_tf/1000)

nim1=nint(i_mc1*tpmc1*2/1000)

nim2=nint(i_mc2*tpmc2*2/1000)

nim3=nint(i_mc3*tpmc3*2/1000)

/title,tfmodb2%rn%, NI(TF/MC1/MC2/MC3):%nit%/%nim1%/%nim2%/%nim3% kA-t

allsel

/edge,1,1

eplo

esel,s,mat,,2

esel,a,mat,,103,105

nsle

eplo

/edge

allsel

save

fini

/solu

tref,0

/com Solve for the conduction problem

allsel

esel,u,type,,1

*if,i_mc1+i_mc2+i_mc3,eq,0,then

esel,u,mat,,103,105  ! Nix MCs

*endif

nsle

solve

/com Solve for the fields

esel,s,mat,,2

*if,i_mc1+i_mc2+i_mc3,ne,0,then

esel,a,mat,,103,105

*endif

nsle

biot,new

allsel

solve

fini

:1000

/post1

/auto

/title,tfmodb2%rn%, NI(TF/MC1/MC2/MC3):%nit%/%nim1%/%nim2%/%nim3% kA-t

esel,s,mat,,2

esel,a,mat,,103,105

nsle

/view,1,1

/edge,1,1

eplo

/edge

/com Modular Coil Fields

esel,s,mat,,103,105

nsle

plns,b,sum

/com TF Coil Fields

esel,s,mat,,2

nsle

plns,b,sum

/com TF Coil Forces

*dim,f_t,,6,ntf/nsym

*do,j,1,ntf/nsym

th%j%=-(360/nsym/2)+arc/2+(360/ntf)*(j-1)

local,1000+j,,,,,th%j%

/psym,csys

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,y,th%j%-arc/2,th%j%+arc/2

esln,,1

rsys,1000+j

etab,fx,fmag,x

etab,fy,fmag,y

etab,fz,fmag,z

csys,1000+j

etab,xc,cent,x 

etab,yc,cent,y  

etab,zc,cent,z  

sadd,eccx,xc,,-1,,r0

/com Calculate Torques from each coil

/com Tx

smult,tx1,fy,zc,-1 

smult,tx2,fz,yc,+1

sadd,tx,tx1,tx2

/com Ty

smult,ty1,fx,zc,+1 

smult,ty2,fz,eccx,+1

sadd,ty,ty1,ty2

/com Tz

smult,tz1,fy,xc,+1 

smult,tz2,fx,yc,-1

sadd,tz,tz1,tz2

ssum

*get,fx,ssum,,item,fx

*get,fy,ssum,,item,fy

*get,fz,ssum,,item,fz

*get,tx,ssum,,item,tx

*get,ty,ssum,,item,ty

*get,tz,ssum,,item,tz

f_t(j,1)=fx

f_t(j,2)=fy

f_t(j,3)=fz

f_t(j,4)=tx

f_t(j,5)=ty

f_t(j,6)=tz

nfx=nint(fx)

nfy=nint(fy)

nfz=nint(fz)

kfx=0.01*nint(fx/10)

kfy=0.01*nint(fy/10)

kfz=0.01*nint(fz/10)

thj=th%j%

/title,tfmodb2%rn%, TF#%j% (%thj% deg), Local FX/FY/FZ=%kfx%/%kfy%/%kfz% kN

/title,tfmodb2%rn%, TF#%j% (%thj% deg), Local FX/FY/FZ=%nfx%/%nfy%/%nfz% N

/view,1,,1

plns,b,sum

*enddo

*vwrite,

('  F(radial)     F(OOP)    F(Vertical)  T(radial)    T(R0)   T(Vertical)')

*vwrite,f_t(1,1),f_t(1,2),f_t(1,3),f_t(1,4),f_t(1,5),f_t(1,6)

(1p6e12.4)

allsel

fini

/exit,nosa

/eof

:2000

srn=2

/filnam,tfmods2%srn%

/show,tfmods2%srn%,grp

/com 21: Contact interface at theta=0 only (others merged), No Radial Preload (missing fcum)

/com 22: Contact interface at theta=0 only (others merged), No Radial Preload (include fcum)

/prep7

shpp,off

dtmp=0

k_nl=1                   ! 0: Wedged surfaces held to UY=0, 1: Generate Flex-Rigid Contact Interface

mu_sw=0.3                ! sidewall-to-sidewall friction coefficient

/com Change to structural elements

et,1,92

et,3,92

/com Delete MC elements

esel,s,type,,2

nsle

edele,all

ndele,all

etdele,2

/com Apply Temps to TF Coil

esel,all

nsle

bfunif,temp,dtmp

/com Apply Wedged Face BCs

*if,k_nl,eq,0,then

/com Couple adjacent nodes and cyclicly couple wedged surfaces at theta edges

csys,1

nsel,s,loc,x,,r0

cpintf,ux

cpintf,uy

cpintf,uz

allsel

nsel,u,loc,y,thmn+th,thmx-th

nrotate,all

/com Cyclic Coupling at inboard legs

cpcyc,ux,0.1*k,1,,thmx-thmn,,1

cpcyc,uy,0.1*k,1,,thmx-thmn,,1

cpcyc,uz,0.1*k,1,,thmx-thmn,,1

*else

/com Generate Node-Node Contact surfaces between Wedged Surfaces for nonlinear analyses

et,11,178,,,1,1,0        ! contact elements

mp,mu,11,mu_sw

mat,11

type,11

*do,j,1,ntf/nsym-1

/gopr

csys,1000+j

wpcsys

wprot,+arc/2

cswpla,2000+j

nsel,s,loc,y,-t,t

csys,1

*get,thmn%j%,node,,mnloc,y

*get,thmx%j%,node,,mxloc,y

th%j%=(thmn%j%+thmx%j%)/2

esel,s,mat,,3

!*if,j,ge,ntf/nsym/2-1,and,j,le,ntf/nsym/2+1,then

*if,j,ge,ntf/nsym/2,and,j,le,ntf/nsym/2,then

real,10+j

r,10+j,,0,2,,,-sin(th%j%)

rmore,cos(th%j%),,2,2      ! 9 & 10 are scale factors on toln & ftol 

eint,,,low

*else

cpintf,ux

cpintf,uy

cpintf,uz

*endif

*enddo

/com Cyclic Coupling at inboard legs

csys,1

nsel,s,loc,x,,r0

allsel

nsel,u,loc,y,thmn+th,thmx-th

nrotate,all

/com Cyclic Coupling at inboard legs

cpcyc,ux,0.1*k,1,,thmx-thmn,,1

cpcyc,uy,0.1*k,1,,thmx-thmn,,1

cpcyc,uz,0.1*k,1,,thmx-thmn,,1

*endif

/com Restrain TF Coil in Z

*do,j,1,ntf/nsym

csys,1000+j

nsel,s,loc,y,-cel/2,cel/2

*if,j,le,ntf/nsym/2,then

*get,zgrnd,node,,mnloc,z

*else

*get,zgrnd,node,,mxloc,z

*endif

nsel,r,loc,z,zgrnd-t,zgrnd+t

*get,xmn_bot,node,,mnloc,x

*get,xmx_bot,node,,mxloc,x

xav_bot=(xmn_bot+xmx_bot)/2

nsel,r,loc,x,xav_bot-cel/2,xav_bot+cel/2

d,all,uz

*enddo

/com Apply Coupling at Equatorial Slit

csys,1

nsel,s,loc,x,r0,2*r0

nsel,r,loc,z,-t,2*t_gap

cpint,ux,2*t_gap

cpint,uy,2*t_gap

cpint,uz,2*t_gap

/com

/com Apply Radial Preload and lateral constraint

/com

f_preload=2000/0.2248    ! applied radial preload per TF Bracket x 2 brackets per coil

z_preload=52*k           ! vertical location of applied preload

*do,j,1,ntf/nsym

csys,1000+j

esel,all

nsel,ext

nsel,r,loc,y,-dz_iwp,dz_iwp

nsel,r,loc,x,r0,12

nsel,r,loc,z,z_preload-cel,z_preload+cel

*get,nnodes,node,,count

nrotate,all

f,all,fx,-f_preload/nnodes

nsle

nsel,ext

nsel,r,loc,y,-dz_iwp,dz_iwp

nsel,r,loc,x,r0,12

nsel,r,loc,z,-z_preload-cel,-z_preload+cel

*get,nnodes,node,,count

nrotate,all

f,all,fx,-f_preload/nnodes

/com Lateral Constraints

allsel

nsel,ext

nsel,r,loc,y,-dz_iwp/2-t,dz_iwp/2+t

nsel,r,loc,x,r0-0.1,r0+0.1

nsel,u,loc,y,-dz_iwp/2+t,dz_iwp/2-t

nrotate,all

d,all,uy

*enddo

allsel

save

fini

/solu

allsel

fcum,add

ldread,forc,2,,,,tfmodb2%rn%,rst

*if,k_nl,eq,0,then

solve

*else

nsubst,25000,50000,10

autots,on

kbc,0

nropt,unsym     ! bombs on "Insufficient memory error during solution" with -m 128/64

!eqslv,pcg  ! PCG solver fails to solve problems with Lagrange multiplier method.

eqslv,sparse

!nropt,full,,off

!lnsrch,on

solve

*endif

fini

:3000

/show,tfmods2%srn%,grp

/post1

/auto

/psym,csys

/cont,,,auto

/title,tfmods2%srn%, Stress Analysis

/view,1,,-1

pldi

esel,s,mat,,2

nsle

plns,s,int

rsys,1

plns,u,x

/com Sidewall Insulation Stresses

*do,j,1,ntf/nsym

csys,1000+j

wpcsys

wprot,-arc/2

cswpla,2000+j

esel,s,mat,,3

nsle

nsel,r,loc,y,-t_sw-t,t_sw+t

esln,,1

rsys,2000+j

etab,sy,s,y

etab,v,volu

etab,sxy,s,xy

etab,syz,s,yz

smult,sxy2,sxy,sxy

smult,syz2,syz,syz

sadd,sxy2syz2,sxy2,syz2

sexp,tau_sw,sxy2syz2,,0.5

sadd,tau_marg,sy,tau_sw,-mu_sw,-1

smult,syv,sy,v

smult,tauv,tau_sw,v

ssum

*get,v_sw,ssum,,item,v

*get,syv,ssum,,item,syv

*get,tauv,ssum,,item,tauv

msy=0.01*nint(syv/v_sw/1e4)

mtau=0.01*nint(tauv/v_sw/1e4)

esel,r,etab,tau_marg,,1e12

ssum

*get,v_oksw,ssum,,item,v

fr_oksw=0.01*nint(100*v_oksw/v_sw)

esln,,1

/title,tfmods2%srn%, Normal Stress on SW #%j%, Average: %msy%MPa

plet,sy,avg

/title,tfmods1%srn%, Shear Stress on SW #%j%, Average: %mtau%MPa

plet,tau_sw,avg

/CONT,1,1,0, ,10e7  

/title,tfmods2%srn%, Frac of Coil%j% Surf Stuck: %fr_oksw%

/view,1,,1

WPSTYLE,,,,,,,,0

plet,tau_marg,avg

/cont,1,9

*enddo

*if,k_nl,eq,1,then

/title,tfmods2%srn%, Relative Motion Across 

esel,s,type,,11

etab,uty,nmisc,6

etab,utz,nmisc,7

smult,uty2,uty,uty

smult,utz2,utz,utz

sadd,uty2utz2,uty2,utz2

sexp,ut,uty2utz2,,.5

plet,ut,avg

*endif

allsel

/cont

fini

/exit,all

/eof

